Technical Document 1598
July 1989



# Specification for Radar Free-Space Detection Range and Free-Space Intercept Range Calculations

C. P. Hattan

# NAVAL OCEAN SYSTEMS CENTER
San Diego, California 92152-5000


E. G. SCHWEIZER, CAPT, USN
Commander

R. M. HILLYER
Technical Director


## ADMINISTRATIVE INFORMATION

The work described in this document was performed from January 1989 to June 1989 by the Tropospheric Branch, Code 543, Naval Ocean Systems Center (NOSC), for NOSC Code 402, Command and Control Department.


Released by
H. V. Hitney, Head
Tropospheric Branch

Under authority of
J. H. Richter, Head
Ocean and Atmospheric
Sciences Division

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE

# REPORT DOCUMENTATION PAGE

| | |
|---|---|
| 1a. REPORT SECURITY CLASSIFICATION<br>UNCLASSIFIED | 1b. RESTRICTIVE MARKINGS |
| 2a. SECURITY CLASSIFICATION AUTHORITY | 3. DISTRIBUTION/AVAILABILITY OF REPORT<br>Approved for public release; distribution is unlimited. |
| 2b. DECLASSIFICATION/DOWNGRADING SCHEDULE | |
| 4. PERFORMING ORGANIZATION REPORT NUMBER(S) | 5. MONITORING ORGANIZATION REPORT NUMBER(S)<br>NOSC TD 1598 |
| 6a. NAME OF PERFORMING ORGANIZATION<br>Naval Ocean Systems Center<br>6b. OFFICE SYMBOL (if applicable)<br>Code 543 | 7a. NAME OF MONITORING ORGANIZATION<br>Naval Ocean Systems Center<br>Material and Device Technology Branch |
| 6c. ADDRESS (City, State and ZIP Code)<br>San Diego, CA 92152-5000 | 7b. ADDRESS (City, State and ZIP Code) |
| 8a. NAME OF FUNDING/SPONSORING ORGANIZATION<br>Naval Oceanographic Office<br>8b. OFFICE SYMBOL (if applicable)<br>Code 9210 | 9. PROCUREMENT INSTRUMENT IDENTIFICATION NUMBER |
| 8c. ADDRESS (City, State and ZIP Code)<br>NSTL Station<br>Bay St. Louis, MS 39522 | 10. SOURCE OF FUNDING NUMBERS<br>PROGRAM ELEMENT NO.: OMN<br>PROJECT NO.:<br>TASK NO.: MP09<br>AGENCY ACCESSION NO.: DN309 022 |

11. TITLE (include Security Classification)

SPECIFICATION FOR RADAR FREE-SPACE DETECTION RANGE AND FREE-SPACE INTERCEPT RANGE CALCULATIONS

12. PERSONAL AUTHOR(S)

C. P. Hattan

| 13a. TYPE OF REPORT | 13b. TIME COVERED | 14. DATE OF REPORT (Year, Month, Day) | 15. PAGE COUNT |
|---|---|---|---|
| Final | FROM Jan 89 TO Jun 89 | July 1989 | 28 |

16. SUPPLEMENTARY NOTATION

| 17. COSATI CODES | | | 18. SUBJECT TERMS (Continue on reverse if necessary and identify by block number) |
|---|---|---|---|
| FIELD | GROUP | SUB-GROUP | radar free-space range<br>ESM free-space range |
| | | | |
| | | | |

19. ABSTRACT (Continue on reverse if necessary and identify by block number)

General-purpose utility programs have been developed to calculate the free-space ranges of a pulse radar and an Electronic Support Measure (ESM) receiver. These utility programs are written in ANSI Fortran. The radar utility program allows the user to calculate radar free-space ranges for arbitrary radar cross-sections, for any probability of detection between 0.1 and 0.9, and for any probability of false alarm between $10^{-12}$ and $10^{-4}$. A similar program to calculate the free-space intercept range for ESM receivers allows the user to input receiver sensitivity, transmitted power, and transmitter antenna gain to obtain a free-space intercept-range value. These programs are subroutines of simple driver programs the user can modify. The programs, as provided, will prompt the user to provide specific input values, and will print a list of the input values as well as the calculated free-space range value at the conclusion of the calculation of that range value. The equivalent one-way free-space path loss for isotropic antennas at the free-space range is also provided. This path loss value can be used with transmission-loss codes or diagrams to assess actual electromagnetic system performance in a real-world environment.

| | |
|---|---|
| 20. DISTRIBUTION/AVAILABILITY OF ABSTRACT<br>☐ UNCLASSIFIED/UNLIMITED ☒ SAME AS RPT ☐ DTIC USERS | 21. ABSTRACT SECURITY CLASSIFICATION<br>UNCLASSIFIED |
| 22a. NAME OF RESPONSIBLE PERSON<br>C. P. Hattan | 22b. TELEPHONE (include Area Code): (619) 553-1427<br>22c. OFFICE SYMBOL: Code 543 |

DD FORM 1473, 84 JAN

83 APR EDITION MAY BE USED UNTIL EXHAUSTED
ALL OTHER EDITIONS ARE OBSOLETE

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE

## CONTENTS



## TABLES



A-1

# 1.0 INTRODUCTION

Free-space range specification is a standard method used to assess electromagnetic (EM) system performance. The specification of performance of a free-space range parameter enables comparisons of different systems even though they are not commonly operated in a "free-space" environment. Free space propagation may be defined as the propagation of energy that would occur if an omnidirectional point-radiating source was placed in outer space. The radiated energy would travel radially outward in all directions, the wave fronts propagating away from the source with the same velocity in all directions. Obviously these conditions would not be satisfied if the point source was placed in the near-earth environment. Refraction by the atmosphere insures that the energy is not propagated with the same velocity in all directions, and the surface of the earth will intercept and reflect some portion of the radiated energy. However, the concept of free-space propagation is useful for systems operated in the terrestrial environment because such propagation can be used as a standard measure for assessing the performance of these systems. The free-space detection range of a radar is often much greater than the actual detection range of that radar against targets near the earth's surface because of horizon effects. Similarly, the free-space intercept range of a sensitive Electronic Support Measure (ESM) receiver may be several hundred, or several thousand, times greater than the actual intercept range under the same conditions.

Utility programs have been developed to calculate the maximum free-space intercept range of an ESM receiver and the maximum free-space detection range of a pulse radar system against a specific-size target. These utilities, written in ANSI Fortran, are designed to be used as simple stand-alone programs, though they can easily be incorporated into other programs if desired.

# 2.0 INPUTS, OUTPUTS, AND LIMITS

## 2.1 INPUTS

The determination of a free-space maximum range requires a knowledge of certain EM system parameters. The number of parameters required depends on the type of system, radar, or ESM. More parameters are required for a radar than for an ESM system. The following paragraphs list the necessary inputs, their units, and their range of validity.

### 2.1.1 Radar System Input Parameters

Table 2-1 lists the required inputs necessary to calculate radar free-space detection range.

**Table 2-1.** Required radar and target parameters.

| Parameter | Units | Input Range |
|---|---|---|
| Simple Radar | n a | Yes or no |
| Coherent Integration | n a | Yes or no |
| Frequency | MHz | 100.0 to 20,000.0 |
| Peak Power | kW | >0.0 to 10,000.0 |
| Pulse Width (or Length) | $\mu s$ | >0.0 to 1,000.0 |
| Pulse Repetition Frequency | Hz | 1.0 to 100,000.0 |
| Receiver Noise Figure | dB | >0.0 to 20.0 |
| Antenna Gain | dB | >0.0 to 50.0 |
| Horizontal Beam Width | deg | >0.0 to 90.0 |
| Horizontal Scan Rate | rpm | 1.0 to 500.0 |
| Hits per Scan | integers | 1.0 to 1000.0 |
| Assumed System Losses | dB | 0.0 to 20.0 |
| Probability of Detection | n a | 0.1 to 0.9 |
| Probabiltity of False Alarm | n a | $10^{-4}$ to $10^{-12}$ |
| Target Radar Cross-Section | $m^2$ | >0.0 to $10^7$ |
| Swerling Case | n a | 0 or 1 |

Some of the parameters listed may not be required, depending on the type of radar for which a maximum free-space range is desired. If the radar is a simple, scanning, two-dimensional (2D) (i.e., range and azimuth) system, then the number of hits per scan is not entered by the user. In this case, the equivalent variable, the number of pulses integrated, is calculated from the horizontal beam width, the antenna horizontal scan rate, and the pulse-repetition-frequency inputs. Conversely, for more complex radar systems, such as height-finder (3D) radars, the number of hits per scan is a program input. In this case, the horizontal beam width, antenna horizontal scan rate, and pulse repetition frequency are not required, and the number of hits per scan must be entered as an integer. A variation of the complex radar is also considered, one which uses coherent integration, the best example of which is a pulse-doppler system. The user is required to specify whether the radar is a simple or complex system and, if complex, whether or not coherent integration is used in processing. The entry for Swerling Case refers to the type of target. The user must also specify either a Swerling Case 0 (steady, nonfluctuating target) or a Swerling Case 1 (slowly fluctuating target).

The assumed-system-loss parameter is used to include all miscellaneous EM system losses. Such losses can include, but are not limited to, collapsing loss, scanning loss, squint loss, pulse-length loss, and system-degradation loss. The system-degradation loss occurs as a result of exposure to weather, poor maintenance or calibration, and deterioration and aging of system components. A value of 5.0 dB, or more, for a shipboard radar is a not uncommon value for such a loss. The other loss mechanisms are described by Blake (1969,1980).

### 2.1.2 ESM Receiver Input Parameters

Table 2-2 lists the EM system inputs required to calculate the free-space intercept range of an ESM receiver. The unit abbreviation dBm, defined as decibels above 1 mW, is a standard way to specify receiver sensitivity threshold levels for ESM systems. The receiver sensitivity includes the gain of the receiving antenna.

**Table 2-2.** Required ESM system inputs.

| Parameter | Units | Input Range |
|---|---|---|
| Frequency | MHz | 100.0 to 20,000.0 |
| Transmitter Power | kW | 0.1 to 10,000.0 |
| Transmitter Antenna Gain | dB | -5.0 to 50.0 |
| Receiver Sensitivity | dBm | -10.0 to -150.0 |

## 2.2 OUTPUTS

The only outputs are the calculated radar or ESM system free-space range values in kilometers and the corresponding one-way free-space path loss at that free-space range. Sample program outputs, and their corresponding inputs, will be given in section 4.0.

## 2.3 LIMITS

The calculation of free-space range described in this document is limited to pulse radars for operational parameters within the range of validity of the inputs of Table 2-1 and ESM systems with operational parameters within the range of validity of the inputs of Table 2-2.

# 3.0 MODELS

There are two different models used to determine free-space EM system range, depending on whether or not the system is a radar or an ESM system. The model for the radar system will be discussed first, followed by a discussion of the model for the ESM system.

## 3.1 RADAR FREE-SPACE RANGE MODEL

Radar free-space range determination is probabilistic, since the detection of a target depends on a great many factors. First, the target must be in the radar beam, and it must be there long enough and with a strong enough return signal that it can be recognized as a target by the detector, whether or not the detector is automatic or human. Second, the detector must be able to discriminate between a target and extraneous background noise. If there are a number of objects at the same range as the target, the return signal from these objects may mask the signal returned from the target. These objects can be natural sources, such as terrain clutter or sea clutter, or man-made sources, such as jammers or chaff. Target return signal is also usually a complex function of aspect angle and may vary rapidly as a function of target motion. Atmospheric refraction, atmospheric noise from extraterrestrial sources, and reflections from the earth's surface, especially the ocean, are examples of other possible natural contributors that can affect the detection process. Given this, it should be apparent that a prediction of maximum radar range cannot be guaranteed to be accurate in a strict sense. Radar performance is often specified in terms of the free-space detection range, which simply ignores the complicating factors which arise from the usual operational environment of the radar system and concentrates on those parts of the detection process that are free of this contamination. This process is still probabilistic, because any return signal will still be mixed in with noise, but it provides a useful reference point to radar system users.

The basic maximum radar range equation, derived from Blake (1980, Eq. 1.28), is given by

$$R_{fs} = 58.0 \left( \frac{G_t^2 \; P_t \; \sigma \; \tau}{fMHz^2 \; N_f \; (S/N)_{\min} \; L} \right)^{1/4} \tag{1}$$

where

| | |
|---|---|
| $R_{fs}$ | = maximum radar free-space detection range in kilometers |
| $P_t$ | = transmitted power in kilowatts |
| $G_t$ | = radar antenna power gain ratio |
| $\sigma$ | = radar target cross-section in square meters |
| $\tau$ | = pulse width (or length) in microseconds |
| $fMHz$ | = radar system frequency in megahertz |
| $(S/N)_{min}$ | = minimum signal-to-noise ratio for a specified probability of detection, probability of false alarms, and Swerling Case 0 or 1 |
| $N_f$ | = receiving system noise figure |
| $L$ | = assumed system losses expressed as a ratio |

This equation assumes that the system is a monostatic radar and that the radar receiver is bandwidth-matched to the pulse width. Equation 1 is applicable to pulse compression radars when $\tau$ is the width of the compressed pulse.

As previously noted, the detection process is statistical or probabilistic, because the returned energy from the target is mixed in with receiver noise. Since this noise is random, there will always be voltage fluctuations at the output of the detector. These fluctuations can make it difficult to determine with any certainty whether or not a target is present. Rapid voltage changes can be due to noise or target return. The probability that a target-generated signal, when present, will be detected is called the *probability of detection, $P_d$*, and the probability that a noise fluctuation will be mistaken for a target is called the *probability of false alarm, $P_{fa}$*. The minimum required signal-to-noise ratio can be defined in terms of these quantities by using an empirically derived formula (Blake, 1980, Eq. 2.29):

$$(S/N)_{\min} = [X_o/(4\, N_p)]\, \{1 + [1 + (16\, N_p/X_o)]^{1/2}\}\, L_f \tag{2}$$

where

$$X_o = (g_{fa} + g_d)^2 \tag{3}$$

$$g_{fa} = 2.36\, (-\operatorname{Log} P_{fa})^{1/2} - 1.02 \tag{4}$$

$$g_d = 1.231\, t\, /(1 - t^2)^{1/2} \tag{5}$$

$$t = 0.9\, (2\, P_d - 1) \tag{6}$$

Equation 2 is sometimes referred to as the detectability factor. $N_p$ is the number of pulses integrated by the detector for a simple 2D scanning system. If the radar is a complex system, height-finder (3D) radar, $N_p$ is the number of hits per scan. $N_p$ for simple radars is given by Blake (1969, Eq. 30):

$$N_p = (\Theta_h\, prf)\, /\, (6\, hsr) \qquad N_p \geq 1.0 \tag{7}$$

where $\Theta_h$ is the horizontal beam width in degrees, *prf* is the pulse repetition frequency in Hertz and *hsr* is the horizontal scan rate in rpm. Unlike the user-entered number of hits per scan, $N_p$ can have noninteger values but is limited to a minimum of a single pulse integrated. $L_f$ is the fluctuation loss ratio ($L_f$ = 1.0 for a steady, Swerling Case 0 target). If the target is fluctuating (Swerling Case 1), the additional loss due to this fluctuation must be calculated. The fluctuation loss ratio, $L_f$, is given by Blake (1980, Eq. 2.45):

$$L_f = [(-\ln P_d)\, (1 + g_d / g_{fa})]^{-1} \tag{8}$$

Equation 2 is valid over the range $0.1 \leq P_d \leq 0.9$ and $10^{-12} \leq P_{fa} \leq 10^{-4}$. In this formulation, a square-law detector, a uniform-weight integrator, and a constant signal power are assumed. Blake (1980, pp. 45–48) notes that square-law detectors are rarely used in radars, but the difference between square-law detectors and linear detectors, which are more commonly used, is generally much less than 1 dB. While Eq. 2 is generally valid for simpler radars that do not use *coherent* integration, Blake (1980, p. 59) notes that the best way to process nonfluctuating or slowly fluctuating multiple-pulse signals is to integrate

them coherently. Such processing, he notes, is achieved only in pulse-doppler radar systems, though some other forms of coherent processing are used, such as moving-target-indicator and pulse-compression systems. When coherent integration is used, Eq. 2 becomes

$$(S/N)_{min} = [X_o/(4 N_p)]\{1 + [1 + (16/X_o)]^{1/2}\} L_f \tag{9}$$

where all quantities are as previously defined. Equation 9 reveals that coherent integration of $N_p$ pulses is equivalent to the detectability factor of a single pulse $N_p$ times as long.

Quantities like the noise figure, $N_f$, antenna gain, $G$, and assumed system losses, $L$, of a radar system are usually stated in decibels. Since Eq. 1 is written with these quantities in terms of ratios, it is convenient to restate Eq. 1 with these terms in decibels. To accomplish this, another quantity is defined which adds all the previously mentioned terms in decibels and allows Eq. 1 to be given in terms of this new quantity:

$$dB_{term} = 10.0^{[2 G_{dB} - N_{fdB} - (S/N)_{dB} - L_{dB}]/10.0} \tag{10}$$

Here the detectability factor from Eq. 2, or Eq. 9, whichever applies, is also included in decibels [10 Log$(S/N)_{min}$]. If this representation is used, Eq. 1 becomes

$$R_{fs} = 58.0 \left( \frac{dB_{term}\, P_t\, \sigma\, \tau}{fMHz^2} \right)^{1/4} \quad \text{(km)} \tag{11}$$

with all other variables retaining their previously defined units. Eq. 11 is the most convenient expression for the calculation of radar free-space range and is the form actually used in the Fortran code.

Equations 2 through 11 are incorporated into a Fortran program, RFSDR (Radar Free-Space Detection Range), consisting of a main routine and three subroutines. When RFSDR is executed, the main routine calls two input subroutines, INRADR and INTRGT, to prompt the operator to enter the required radar and target parameters of Table 2-1. Upon return from the INRADR and INTRGT subroutines, the free-space range is calculated from Eq. 11. The equivalent one-way path loss at the free-space range, assuming isotropic antennas, is also calculated at this time by using the basic transmission equation from Kerr (1951, Eq. 2-15):

$$\text{Loss} = 32.45 + 20.0\ \text{Log}\,(R_{fs}\, fMHz) \tag{12}$$

The free-space loss value is useful for evaluating radar system performance with transmission-loss plots for actual operational environments. The third subroutine, RDROUT, is then called to display a list of the input parameters and the calculated free-space range. Program operation is terminated after the radar and target inputs, the calculated free-space range, and the path-loss values are listed. Complete program listings for RFSDR are included in Appendix A.

To use RFSDR, the operator must compile and link the Fortran routines listed in Appendix A. Numerical radar system or target inputs outside the specified input range will be program-limited to the nearest bound specified in Table 2-1. That is, if a frequency of 50 MHz is entered, the program will automatically set the frequency variable to 100 MHz, the lower frequency bound. Blank field entries are not acceptable numerical inputs. Those inputs listed in Table 2-1 that require a numerical value greater than zero will be defaulted to 0.1 if the user inputs a negative or zero value. This is to prevent runtime errors at

program execution. Non-numeric (alpha-character) inputs, such as yes no responses (e.g., type of system), will accept any character, even blanks. The default for such an entry is the first option listed in the operator prompt. An example would be the choice of the type of radar system. The INTRGT operator prompt asks: "Is the radar a simple system [range and azimuth only, nonheight finder, simple signal processing] (y or n)?" Any response, other than an "n," will cause the system to consider the input as a simple radar system. Program operation is as stated in the previous paragraph.

## 3.2 ESM RECEIVER FREE-SPACE RANCE MODEL

The free-space range equation for an ESM receiver, derived from Kerr (1951, Eq. 2-15),is given by

$$R_{fs} = 10.0^{(10\,\mathrm{Log}(Pt) - 20\,\mathrm{Log}(fMHz) + Gt - S + 27.5517)/20.0}\ (\mathrm{km}) \tag{13}$$

where

$R_{fs}$ = free-space intercept range in kilometers

$P_t$ = the transmitted power in kilowatts

$G_t$ = gain in dB above an isotropic radiator of the transmitter antenna

$fMHz$ = frequency in megahertz

$S$ = sensitivity of the receiver, including receiving antenna gain, in dBm

The range values obtained by using Eq. 13 can be extremely large, perhaps tens of thousands of kilometers. The free-space range can be stated as an equivalent free-space loss value for isotropic antennas by using Eq. 12. This one-way path-loss value can be used to evaluate actual ESM receiver intercept-range performance when used with transmission-loss plots or EM propagation codes. The ESM free-space range program returns both the range from Eq. 13 and the equivalent free-space path loss for isotropic antennas from Eq. 12.

Equations 13 and 12 are incorporated into a Fortran program, FSIR (Free-Space Intercept Range), consisting of a main routine and two subroutines. When FSIR is executed, the main routine calls INPUTS to prompt the operator to enter the necessary EM system parameters of Table 2-2. Upon return from INPUTS, the main routine calculates the free-space range and the one-way path loss at the free-space range by using Eq. 13 and 12, respectively. The main routine then calls ESMOUT. This subroutine is used to display the input parameters and the calculated free-space range and path-loss values. Program execution is then terminated. Complete program listings for FSIR are included in Appendix B.

To use FSIR, the operator must compile and link the Fortran routines listed in Appendix B. Numerical EM system inputs outside the specified input range will be program-limited to the nearest bound specified in Table 2-2. That is, if a frequency of 21,000 MHz is entered, the program will automatically set the frequency variable to 20,000 MHz, the upper frequency bound. Blank field entries are not acceptable numerical inputs. The program operational sequence is detailed in the preceding paragraph.

## 4.0 TEST CASES

This section describes test and evaluation criteria for the RFSDR and FSIR programs. The test cases will allow program users to verify correct operation.

### 4.1 RFSDR TEST CASES

Table 4-1 lists five test cases that should verify that the RFSDR program is operating correctly. The input and output parameters are listed to the nearest tenth of a unit. RFSDR is operating correctly if the free-space range is within 0.1 km of the value listed in Table 4-1. The abbreviations for the variables are the same as those listed in section 3.1, with the following exceptions. The radar type is listed as either an "S" (simple 2D system) or a "C" (complex system). If the radar is not a simple system it may use coherent processing. In this case the line labeled "Coherent" will have a yes or no entry. The target Swerling Case is given in the heading labeled "SW Case." The one-way free-space path loss from Eq. 15 is given in the heading labeled "Loss." An entry of "n a" indicates that this variable is not used in the test case.

**Table 4-1.** RFSDR test case inputs and outputs.

| Parameter | Test Case | | | | |
|---|---|---|---|---|---|
| | 1 | 2 | 3 | 4 | 5 |
| Radar | S | S | S | C | C |
| Coherent | n a | n a | n a | No | Yes |
| $f$ $MHz$ | 100.0 | 20,000.0 | 100.0 | 5000.0 | 10,000.0 |
| $P_t$ ($kW$) | 1.0 | 10,000.0 | 10,000.0 | 5000.0 | 4000.0 |
| $\tau$ ($\mu s$) | 0.1 | 1000.0 | 1000.0 | 100.0 | 1000.0 |
| $N_f$ (dB) | 0.1 | 20.0 | 20.0 | 20.0 | 20.0 |
| $G$ (dB) | 0.1 | 50.0 | 50.0 | 30.0 | 35.0 |
| $L$ (dB) | 0.0 | 0.0 | 0.0 | 10.0 | 5.0 |
| $Prf$ (Hz) | 1.0 | 100,000.0 | 100,000.0 | n a | n a |
| $\Theta_h$ (deg) | 1.0 | 90.0 | 20.0 | n a | n a |
| $Hsr$ (rpm) | 1.0 | 500.0 | 500.0 | n a | n a |
| Hits Scan | n a | n a | n a | 80.0 | 100.0 |
| $\sigma$ ($m^2$) | 1.0 | 1.0 | 1.0 | $10^7$ | 1.0 |
| $P_d$ | 0.9 | 0.9 | 0.9 | 0.1 | 0.9 |
| $P_{fa}$ | $10^{12}$ | $10^{4}$ | $10^{12}$ | $10^{12}$ | $10^{4}$ |
| SW Case | 1 | 0 | 0 | 0 | 1 |
| $R_{fs}$ (km) | [illegible] | 4176.8 | 42274.0 | 7353.7 | 346.1 |
| Loss (dB) | [illegible] | 190.9 | 165.0 | 183.8 | 163.2 |

### 4.2 FSIR TEST CASES

Table 4-2 lists five test cases that should verify that the FSIR program is operating correctly. The input and output parameters are listed to the nearest tenth of a unit. FSIR is operating correctly if the free-space range is within 0.1 km of the value listed in Table 4-2 for test cases 1, 2, 3, and 5. For test case 4, the number of significant digits precludes using the free-space range. In this test, the one-way path loss should be within 0.1 dB. The abbreviations for the variables are the same as those listed in section 3.2.

**Table 4-2.** FSIR test case inputs and outputs.

| Parameter | Test Case | | | | |
|---|---|---|---|---|---|
| | 1 | 2 | 3 | 4 | 5 |
| $f$ *MHz* | 100.0 | 100.0 | 20,000.0 | 20,000.0 | 5500.0 |
| $P_t$ (kW) | 0.1 | 10,000.0 | 10,000.0 | 10,000.0 | 1000.0 |
| $G_t$ (dB) | 5.0 | 50.0 | 5.0 | 50.0 | 35.0 |
| $S$ (dBm) | 10.0 | 10.0 | 150.0 | 150.0 | -90.0 |
| $R_{fs}$ (km) | 0.4 | 23,855.3 | 2,121,068.8 | $1.193e^9$ | 243,905.9 |
| Loss (dB) | 65.0 | 160.0 | 245.0 | 300.0 | 215.0 |

## 5.0 REFERENCES


Blake, L.V., "A Guide to Basic Pulse-Radar Maximum-Range Calculation - Part I - Equations, Definitions, and Aids to Calculation," Washington, D.C., Naval Research Laboratory, NRL Report 6930, 23 December, 1969.

Blake, L.V., *Radar Range-Performance Analysis*, Lexington, MA, Lexington Books, 1980.

Kerr, D.E., *Propagation of Short Radio Waves*, New York, McGraw-Hill, 1951, p. 31.

## Appendix A
## RADAR FREE-SPACE RANGE (RFSDR) PROGRAM LISTS

```
c
c     RFSDR calculates the free-space range of a radar and the pathloss
c     at the free-space range.
c
c     VARIABLES:              DESCRIPTION:
c     antgn          Radar antenna gain in dB
c     dbterm         Various system gains and losses in dB
c     floss          Fluctuating target loss for Swerling Case 1 Target
c     fmhz           Radar frequency in MHz
c     hitscn         Hits per scan, specified for 3D radars and
c                    calculated for 2D radars as the number of
c                    pulses integrated
c     hbwidth        Horizontal beam width in degrees
c     hscnr8         Horizontal scan rate in rpm
c     icoher         Coherent integration flag [1 = yes, 0 = no]
c     icmplx         Complex system (3D radar) flag [1 = yes, 0 = no]
c     itgt           Fluctuating target flag [1 = fluct., 0 = steady]
c     pkpwr          Peak radar power in kw
c     plsr8          Pulse repetition frequency (or rate) in Hz
c     psubd          Probability of detection
c     psubfa         Probability of false alarms
c     pthlos         Pathloss at radar free-space range in dB
c     rcvrnf         Radar receiver noise figure in dB
c     rsubfs         Radar free-space range in km
c     sigma          Radar cross-section of target in square meters
c     snmin          Minimum signal-to-noise ratio in dB
c     syslos         Assumed radar system losses in dB
c     tau            Pulse width (or length) in microseconds
c
c     Call subroutine incalc to enter radar parameters.
c
      real*4 antgn,dbterm,floss,fmhz,gsubd,gsubfa,hitscn,hbwidth,hscnr8
      real*4 pkpwr,plsr8,pthlos,psubd,pusbfa,rcvrnf,rsubfs,sigma,snmin
      real*4 syslos,t,tau,Xsub0
      integer*2 icoher,icmplx,itgt
c
c     Call subroutine inradr to enter radar parameters.
c
      call inradr(antgn,fmhz,hitscn,hbwidth,hscnr8,icoher,icmplx,
     1            pkpwr,plsr8,rcvrnf,syslos,tau)
c
c     Call subroutine intrgt to enter target parameters.
c
      call intrgt(itgt,psubd,psubfa,sigma)
c
      T = .9*(2.*psubd - 1.)
      gsubd = 1.231*T / SQRT(1.0 - T**2)
      gsubfa = 2.36 * SQRT(-LOG10(psubfa)) - 1.02
      Xsub0 = (gsubfa + gsubd)**2
      IF ((icoher .EQ. 1) .AND. (icmplx .EQ. 1)) THEN
```

```
c          If coherent processing is used [pulse doppler radar] icoher=1
         snmin = Xsub0/4.0/hitscn*(1.0 + SQRT(1.0 + 16.0/Xsub0))
        ELSE
c          Incoherent integration.
         snmin = Xsub0/4.0/hitscn*(1.0 + SQRT(1.0 + 16.0*hitscn/Xsub0))
        END IF
        floss = 1.0
c       If target type is 'fluctuating' calculate fluctuation loss ratio.
        IF (itgt .EQ. 1)floss = ((-LOG(psubd))*(1.0 + gsubd/gsubfa))**(-1)
        snmin = snmin * floss

        sndb = 10.0 * LOG10(snmin)
        dbterm = 10.0**((2.0*antgn-rcvrnf-sndb-syslos)/10.0)
        rsubfs = 58.0 * (dbterm*pkpwr*sigma*tau/fmhz**2)**.25
        pthlos = 32.45 + 20.0 * LOG10(rsubfs * fmhz)
c
c       Call subroutine prtout to print out input values and calculated
c       free-space range and pathloss.
c
        call rdrout(antgn,fmhz,hitscn,hbwidth,hscnr8,icoher,icmplx,itgt,
     1             pkpwr,plsr8,psubd,psubfa,rcvrnf,sigma,syslos,tau,
     2             rsubfs,pthlos)
c       RETURN
        END
```

```
c     Subroutine INRADR prompts the user for radar system parameters
c     and returns to the free-space range subroutine FSRNG to deter-
c     mine the free-space range of the radar for the specified in-
c     puts and the pathloss at the free-space range.
c
c
      subroutine inradr(antgn,fmhz,hitscn,hbwidth,hscnr8,icoher,icmplx,
     1                  pkpwr,plsr8,rcvrnf,syslos,tau)
c
      real*4 antgn,fmhz,hitscn,hbwidth,hscnr8,pkpwr,plsr8,tau
      real*4 rcvrnf,syslos
      character*1 dummy
      integer*2 icoher,icmplx,ZW,ZR
c
c       Specify the read (5) and write (6) channel numbers.
      ZW = 6
      ZR = 5
c       Enter the radar system parameters.
      write(ZW,'("Enter Radar System Parameters ")')
c       Set the radar type to simple radar [ icmplx = 0 ].
      icmplx = 0
      write(ZW,1000)
 1000 format('Is the radar a simple system [range and azimuth only,',
     1 /,'non-height-finder, simple signal processing] (y or n)? ',$)
      read(ZR,'(A1)')dummy
      IF ( dummy .EQ. 'n' ) icmplx = 1
c       Set the integrator type to non-coherent [ icoher = 0 ].
      icoher = 0
      IF ( icmplx .EQ. 1 ) THEN
        write(ZW,1005)
 1005   format('Does the radar use non-coherent integration or',
     1         /,'coherent integration [pulse doppler radar] '
     2         '(n or c)? ',$)
        read(ZR,'(A1)')dummy
        IF (dummy .EQ. 'c') icoher = 1
      END IF
c
      write(ZW,1010)
 1010 format('Enter frequency in MHz (100 to 20,000) ',$)
      read(ZR,*) fmhz
      IF (fmhz .LT. 100.0) fmhz = 100.0
      IF (fmhz .GT. 20000.0) fmhz = 20000.0
c
      write(ZW,1015)
 1015 format('Enter peak power in kw (1.0 to 10,000) ',$)
      read(ZR,*) pkpwr
      IF (pkpwr .LT. 1.0) pkpwr = 1.0
      IF (pkpwr .GT. 10000.0) pkpwr = 10000.0
c
      write(ZW,1020)
 1020 format('Enter pulse width in microseconds ',
     1       '(>0. to 1000.0) ',$)
      read(ZR,*) tau
      IF (tau .LE. 0.0) tau = 0.1
      IF (tau .GT. 1000.0) tau = 1000.0
```

```
c
      write(ZW,1025)
 1025 format('Enter receiver noise figure in dB (>0. to 20.0) ',$)
      read(ZR,*) rcvrnf

      IF (rcvrnf .LE. 0.0) rcvrnf = 0.1
      IF (rcvrnf .GT. 20.0) rcvrnf = 20.0
c
      write(ZW,1030)
 1030 format('Enter antenna gain in dB (>0. to 50.0) ',$)
      read(ZR,*) antgn
      IF (antgn .LE. 0.0) antgn = 0.1
      IF (antgn .GT. 50.0) antgn = 50.0
c
      write(ZW,1035)
 1035 format('Enter assumed system losses in dB (0.0 to 20.0) ',$)
      read(ZR,*) syslos
      IF (syslos .LE. 0.0) syslos = 0.0
      IF (syslos .GT. 20.0) syslos = 20.0
c
      IF (icmplx .EQ. 1) THEN
        write(ZW,1040)
 1040   format('Enter the number of hits per scan (1 to 1000) ',$)
        read(ZR,*) hitscn
        IF (hitscn .LT. 1.0) hitscn = 1.0
        IF (hitscn .GT. 1000.0) hitscn = 1000.0
      ELSE
c
        write(ZW,1045)
 1045   format('Enter pulse repetition frequency in Hz ',
     1          '(1.0 to 100,000.0) ',$)
        read(ZR,*) plsr8
        IF (plsr8 .LE. 0.0) plsr8 = 0.1
        IF (plsr8 .GT. 1.0E5) plsr8 = 1.0E5
c
        write(ZW,1050)
 1050   format('Enter horizontal beam width in ',
     1          'degrees (>0.0 to 90.0) ',$)
        read(ZR,*) hbwidth
        IF (hbwidth .LE. 0.0) hbwidth = 0.1
        IF (hbwidth .GT. 90.0) hbwidth = 90.0
c
        write(ZW,1055)
 1055   format('Enter horizontal scan rate ',
     1          'in rpm (1.0 to 500.0) ',$)
        read(ZR,*) hscnr8
        IF (hscnr8 .LT. 1.0) hscnr8 = 1.0
        IF (hscnr8 .GT. 500.0) hscnr8 = 500.0
```

```
c
c       Calculate the number of pulses integrated from the horiz. beam
c       width, horiz. scan rate and the pulse repetition rate.
c
        hitscn = hbwidth*plsr8/6.0/hscnr8
        IF (hitscn .LT. 1.0) hitscn = 1.0
      END IF
c
      RETURN
      END
```

```
c
c     Subroutine INTRGT prompts the user for target parameters:
c     target size, probability of detection, probability of false
c     alarms and if the target is fluctuating or steady. It then
c     returns to the free-space range subroutine fsrng to deter-
c     mine the free-space range of the radar for the specified in-
c     puts and the pathloss at the free-space range.
c
c
      subroutine intrgt(itgt,psubd,psubfa,sigma)
c
      real*4 pexp,psubd,psubfa,sigma
      character*1 dummy
      integer*2 iexp,itgt,ZW,ZR
c
c       Specify the read (5) and write (6) channel numbers.
      ZW = 6
      ZR = 5
c
      write(ZW,1000)
 1000 format('Enter target radar cross section in square meters',
     1       ' (>0.0 to 1.e+07) ',$)
      read(ZR,*) sigma
      IF (sigma .LE. 0.0) sigma = 0.10
      IF (sigma .GT. 1.0E7) sigma = 1.0E7
c
      write(ZW,1005)
 1005 format('Enter probability of detection (.1 to .9) ',$)
      read(ZR,*) Psubd
      IF (Psubd .GT. 0.9) Psubd = 0.9
      IF (Psubd .LT. 0.1) Psubd = 0.1
c
      write(ZW,1010)
 1010 format('Enter probability of false alarms (4 to 12) 1.0e-',$)
      read(ZR,*) pexp
      IF(pexp .LT. 4.0)pexp = 4.0
      IF(pexp .GT. 12.0)pexp = 12.0
      iexp = INT(pexp)
      Psubfa = 10.0**(-iexp)
c
c       Set the target type to 'fluctuating' [itgt = 1].
      itgt = 1
      write(ZW,1015)
 1015 format('Is the target fluctuating [Swerling Case 1] or '
     1         /,19x,'steady [Swerling Case 0] (1 or 0)? ',$)
      read(ZR,'(A1)')dummy
      IF (dummy .EQ. '0') itgt = 0
c
      RETURN
      END
```

```
      subroutine rdrout(antgn,fmhz,hitscn,hbwidth,hscnr8,icoher,icmplx,
     1                  itgt,pkpwr,plsr8,psubd,psubfa,rcvrnf,sigma,
     2                  syslos,tau,rsubfs,pthlos)
c
c     Subroutine RDROUT provides the user with the radar system para-
c     meter inputs, the target size, probability of detection and false
c     alarm rate and the calculated free-space range and path-loss values
c     from subroutine RFSDR.
c
c
      real*4 antgn,fmhz,hitscn,hbwidth,hscnr8,pkpwr,plsr8
      real*4 psubd,psubfa,pthlos,rcvrnf,rsubfs,sigma,syslos,tau
      integer*2 icmplx,icoher,ZW,ZR
c
c       Specify the read (5) and write (6) channel numbers.
      ZW = 6
      ZR = 5
c
      IF ( icmplx .EQ. 0) THEN
        write(ZW,1000)
 1000   format(///'Radar is a simple 2D system (range & azimuth) ')
      ELSE
        write(ZW,1005)
 1005   format(///'Radar is not a simple 2D system ')
      END IF
c
      IF ( icoher .EQ. 1 ) THEN
        write(ZW,1010)
 1010   format('Radar uses coherent integration [pulse doppler radar]')
      END IF
c
      write(ZW,1015) fmhz
 1015 format('Radar frequency = ',f8.1,' MHz')
c
      write(ZW,1020) pkpwr
 1020 format('Radar peak power = ',f8.1,' kw')
c
      write(ZW,1025) tau
 1025 format('Radar pulse width = ',f6.1,' microseconds ')
c
      write(ZW,1030) rcvrnf
 1030 format('Radar receiver noise figure = ',f4.1,' dB')
c
      write(ZW,1035) antgn
 1035 format('Radar antenna gain = ',f4.1,' dB')
c
      write(ZW,1040) syslos
 1040 format('Assumed radar system losses  = ',f4.1,' dB')
c
      IF (icmplx .EQ. 1) THEN
        write(ZW,1045) hitscn
 1045   format('Number of hits per scan = ',f6.1)
      ELSE
```

```
      write(ZW,1050) plsr8
 1050   format('Radar pulse repetition frequency = ',f9.1,' Hz')
c
        write(ZW,1055) hbwidth
 1055   format('Horizontal beam width = ',f4.1,' degrees')
c
        write(ZW,1060) hscnr8
 1060   format('Horizontal scan rate = ',f5.1,' rpm')
c
      END IF
c
      IF ( itgt .EQ. 0 ) THEN
        write(ZW,'("Target is steady, non-fluctuating, "
     1             "Swerling Case 0")')
      ELSE
        write(ZW,'("Target is fluctuating, Swerling Case 1")')
      END IF
c
      write(ZW,1070) sigma
 1070 format('Target radar cross section = ',pe7.1,' square meters')
c
      write(ZW,1075) psubd
 1075 format('Probability of detection = ',f3.1)
c
      write(ZW,1080) psubfa
 1080 format('Probability of false alarms = ',pe8.1)
c
      write(ZW,1085) rsubfs
 1085 format(//,'Radar free-space range = ',f12.1,' km')
c
      write(ZW,1090) pthlos
 1090 format('One-way path loss at the radar free-space range = ',
     1        f12.1,' dB',4(/))
      RETURN
      END
```

## Appendix B
## FREE-SPACE INTERCEPT RANGE (FSIR) PROGRAM LISTS

```
c
c           FSIR:  Calculates free-space ESM intercept range.
c
c     VARIABLES:          DESCRIPTION:
c
c     antgn               Transmitting antenna gain in dB
c     esmrng              ESM intercept range
c     fmhz                Frequency in MHz
c     fsloss              Free-space path loss at esmrng, in dB
c     logtrm              Temporary variable - log of powrkw & fmhz terms
c     powrkw              Transmitter output power in kW
c     rsens               ESM receiver sensitivity in dBm
c
      real*4 antgn, esmrng, fmhz, fsloss, logtrm, powrkw, rsens
c
c          Call inputs to enter EM system parameters
c
      CALL inputs(antgn,fmhz,powrkw,rsens)
c
c
      logtrm = 10.0 * ALOG10( powrkw / (fmhz * fmhz) )
c
      esmrng = 10.0**(( logtrm + antgn - rsens + 27.5517 ) / 20.0 )
c
      fsloss = 20.0 * ALOG10( fmhz * esmrng ) + 32.45
c
c           Call esmout to print out calculated values and inputs
c
      CALL esmout(antgn,esmrng,fmhz,fsloss,powrkw,rsens)
c
      stop
      end
```

```
      subroutine INPUTS(antgn, fmhz, powrkw, rsens)
c
c     Subroutine to enter the necessary inputs to determine the free-
c     space intercept range for an ESM receiver.
c
c     VARIABLE NAMES:             VARIABLE DESCRIPTION:
c
c     antgn                       Transmitting antenna gain in dB.
c     fmhz                        EM system frequency in MHz.
c     powrkw                      Transmitted power in KW.
c     rsens                       Receiver sensitivity in dBm.
c     ZR                          Read channel number
c     ZW                          Write channel number
c
c
      REAL*4 antgn, fmhz, powrkw, rsens
      INTEGER*2 ZW,ZR
c
      ZR = 5
      ZW = 6
c
      WRITE (ZW,1100)
 1100 FORMAT ("This program will calculate ESM free space")
      WRITE (ZW,1105)
 1105 FORMAT ("intercept range and path loss threshold.")
c
      WRITE (ZW,1200)
 1200 FORMAT (/,'Enter frequency in MHz (100 to 20000) ',$)
      READ ( ZR, * ) fmhz
      IF ( fmhz .LT. 100.0 ) fmhz = 100.0
      IF ( fmhz .GT. 20000.0 ) fmhz = 20000.0
c
      WRITE (ZW,1300)
 1300 FORMAT (/,'Enter transmitter power in kW (0.1 to 10000) ',$)
      READ ( ZR, * ) powrkw
      IF ( powrkw .LT. 0.1) powrkw = 0.10
      IF ( powrkw .GT. 10000.0 ) powrkw = 10000.0
c
      WRITE (ZW,1400)
 1400 FORMAT (/,'Enter transmitter antenna gain in dB (-5 to 50) ',$)
      READ ( ZR, * ) antgn
      IF ( antgn .LT. -5.0 ) antgn = -5.0
      IF ( antgn .GT. 50.0 ) antgn = 50.0
c
      WRITE (ZW,1500)
 1500 FORMAT (/,'Enter receiver sensitivity in dBm (-10 to -150) ',$)
      READ ( ZR, * ) rsens
      IF ( rsens .LT. -150.0 ) rsens = -150.0
      IF ( rsens .GT. -10.0 ) rsens = -10.0
c
c
      return
      end
```

```
      subroutine ESMOUT(antgn, esmrng, fmhz, fsloss, powrkw, rsens)
c
c     Subroutine to enter the necessary inputs to determine the free-
c     space intercept range for an ESM receiver.
c
c     VARIABLE NAMES:            VARIABLE DESCRIPTION:
c
c     antgn                      Transmitting antenna gain in dB.
c     esmrng                     ESM intercept range in km.
c     fmhz                       EM system frequency in MHz.
c     fsloss                     Free-space path loss at esmrng, in dB.
c     powrkw                     Transmitted power in KW.
c     rsens                      Receiver sensitivity in dBm.
c     ZW                         Write channel number
c
c
      REAL*4 antgn, esmrng, fmhz, fsloss, powrkw, rsens
      INTEGER*2 ZW
c
c     Set write channel
      ZW = 6
c
      WRITE (ZW,1000) fmhz
 1000 FORMAT (////,'Frequency = ',f8.1,' MHz')
c
      WRITE (ZW,1100) powrkw
 1100 FORMAT ('Transmitter power = ',f8.1,' kw')
c
      WRITE (ZW,1200) antgn
 1200 FORMAT ('Transmitter antenna gain = ',f5.1,' dB ')
c
      WRITE (ZW,1300) rsens
 1300 FORMAT ('Receiver sensitivity = ',f7.1,' dBm ')
c
      IF (esmrng .LE. 1.0e8) THEN
        write (ZW,1400) esmrng
 1400   FORMAT (/ ,'Free space range = ',F11.1,' km')
      ELSE
        write (ZW,1500) esmrng
 1500   FORMAT (/ ,'Free space range = ',1pe14.7,' km')
      END IF
c
      write (ZW,1600) fsloss
 1600 FORMAT ('Path loss threshold = ',F6.1,' dB',5(/))
      return
      end
```

## Initial Distribution

Naval Oceanographic Office
Stennis Space Center, MS 39522-5000
Code 9200

Naval Environmental Prediction Research
Facility
Monterey, CA 93943-5006

Center for Naval Analyses
Alexandria, VA 22302-0268

NOSC Washington Liaison Office
Washington, DC 20363-5100

Defense Technical Information Center
Alexandria, VA 22304-6145